
Abstract

In this paper we compare the formal specification languages Z and Ab-
stract Machine Notation (AMN); the latter of which is due to Abrial.
The strategy adopted is that of presenting the same specification both in
Z and AMN and of commenting on salient differences as they arise. The
specification chosen is a slightly revised version of the specification of an
Internal Telephone Number Database found in chapter 4 of [4]. At the
end of the paper some general conclusions are drawn.

1 Introduction

Z is a well-established, formal specification language. It was developed about
15 years ago, has had many books and articles written about it and there is
a sizable community of users both in academia and also in industry. Abrial’s
Abstract Machine Notation (AMN) is, by contrast, a recent innovation with
only a handful of users and little has been published about it. One of the aims of
this paper is to introduce the person familiar with Z to AMN and to compare
these two specification languages. In such a short paper it is impossible to
mention every interesting feature of AMN, but we hope to convey the “feel” of
a specification written in AMN.

Our goal in writing this paper is not primarily to try and convince Z users
to convert to being users of AMN. However, in addition to introducing AMN to
a wider audience we also hope to add to the current debate in the Z community
about whether or not object-oriented features should be incorporated into Z
and if they should how this ought to be done. A study of AMN can help
someone appreciate the issues involved as AMN is built around the equivalent
in the realm of specification of what in the realm of programming language
theory is known as a module or a class or a package.

The approach taken in this paper is to present a specification of a small
system both in Z and in AMN and to comment on some of the important

research and for that we are very grateful.

differences between them. The system chosen for this purpose is that of an
Internal Telephone Number Database found in chapter 4 of [4], though slight
changes have been made to the specification found there.

2 The Internal Telephone Number Database

In this section we give an informal description in English of the functionality
of the system to be specified:1

An organisation wants to computerise its internal telephone direc-
tory. The database must keep a record of all the people who are
currently members of the organisation (as only they can have tele-
phone extensions). The database must cope with the possibility
that one person may be reached at several extensions and also with
the possibility that several people might have to share an extension.
Six operations are to be provided, namely:

(1) Adding a person to the database’s record of who is currently
a member of the organisation.

(2) Removing a person from the database.

(3) Adding an entry to the database, where an entry is an associ-
ation between a person and a telephone extension.

(4) Removing an entry from the database.

(5) Interrogating the database by person.

(6) Interrogating the database by extension number.

3 Specification Prologue and Initialisation

In this paper the prologue of a specification refers to that part of it which
excludes the operations. The prologue, so to speak, sets the scene for the
specification of the operations that we are interested in. First, the prologue
will be given in Z and then it will be presented using Abrial’s AMN.2

The specification of the telephone directory makes use of two basic types,
which in Z are introduced by means of a basic type definition as follows:

[PEOPLE,PHONES].

PEOPLE is the type of all people and PHONES is the type of all possible
telephone numbers. We also need a set of messages. This is called REPORTS

1This is based on the informal account that is found near the beginning of chapter 4 of
[4].

2In [1], pp. 3–6, what we call the prologue is referred to as the statics. Furthermore, Abrial
calls the definition of the operations in an abstract machine the dynamics. This terminology
is not used of Z specifications, so we decided to use the neutral term prologue.

and it can be defined by means of a free type definition like this:

REPORTS ::= Okay
| Already Member
| Not Member
| Entry Already Exists
| Unknown Entry
| Unknown Name
| Unknown Number.

The state of the internal telephone number database is given by means of the
schema PhoneDB:

PhoneDB
members : P(PEOPLE)
telephones : PEOPLE ↔ PHONES

dom(telephones) ⊆ members

The set members consists of all the members of the organisation and the re-
lation telephones records the information of who can be reached at which ex-
tension. An element of the relation telephones will be called an entry and an
example of such an entry is the ordered pair or maplet diller 7→ 4794, which
tells us that diller can be reached at the extension whose number is 4794. The
state invariant of the schema PhoneDB is the single formula:

dom(telephones) ⊆ members.

This tells us that everyone who has a telephone extension must be a member
of the organisation.

Although not part of the prologue, it is convenient to mention the schemas
∆PhoneDB, ΞPhoneDB and InitPhoneDB′ here. The schema ∆PhoneDB
is a convenient way of referring to all the variables needed to specify an oper-
ation. It is defined as follows:

∆PhoneDB =̂ PhoneDB ∧ PhoneDB′.

The schema ΞPhoneDB is used in the specification of operations that do not
change the state of the database:

ΞPhoneDB =̂ [∆PhoneDB|members′ = members ∧
telephones′ = telephones].

The initial state of the system is InitPhoneDB′, where InitPhoneDB is de-
fined in this way:

InitPhoneDB
PhoneDB

members = ∅
telephones = ∅

A Z specification can be thought of as defining an abstract data type. However,
as [10], p. 129, puts it, ‘For a specification to describe a genuine abstract data
type, there must be at least one possible initial state.’ In the case of the
telephone number database, the requirement that there be at least one possible
initial state is satisfied if the following can be proved to hold:

∃PhoneDB′ • InitPhoneDB′.

Written out in full this formula looks like this:

∃members′ : P(PEOPLE); telephones′ : PEOPLE ↔ PHONES|
dom(telephones′) ⊆ members′ •members′ = ∅ ∧ telephones′ = ∅.

This is true as it simplifies to dom(∅) ⊆ ∅.
The specification prologue in AMN should be readily understood by anyone

familiar with Z. The specification is as follows:

machine

Phone

sets

PEOPLE;PHONES;REPORTS = { Okay, . . . }
variables

members; telephones

invariant

members ⊆ PEOPLE ∧
telephones ∈ PEOPLE ↔ PHONES ∧
dom(telephones) ⊆ members

initialisation

members, telephones := ∅, ∅
operations

...

end

One of the major differences between Z and AMN is that in AMN the en-
tire specification of a single software system is packaged up into an abstract
machine—which may well make use of other machines—that is a well-defined
entity in its own right. In the above (incomplete) machine, the prologue consists
of everything apart from what occurs in the initialisation and operations
components. What goes into the operations component in order to complete
the machine will be explained in section 6 below. According to [1], p. 1, the
notion of an abstract machine is one of the central features of AMN and it
is the analogue in the context of specification of what—in the context of pro-
gramming language description—is known as ‘a class (SIMULA), or a module
(MODULA), or a package (ADA)’.3 A minor difference between Z and AMN
can be discerned here and it is that AMN does not use the typed predicate
calculus which Z uses.

3‘Ada’ is a registered trademark of the US Government (Ada Joint Program Office).

The name of the abstract machine displayed above is Phone. In the sets
component of the machine are found both the basic types of the corresponding
Z specification, namely PEOPLE and PHONES. The collection of mes-
sages, namely REPORTS, is also placed there. What occurs in the variables
component of the machine corresponds to the variables declared in the state
schema PhoneDB. The state invariant from that schema is found in the in-
variant component of the machine, where any other information, such as type
statements, constraining the values of the variables members and telephones
also occurs. It should be noted that the symbol := that occurs in the ini-
tialisation component of the machine is not assignment but substitution and
members, telephones := ∅, ∅ refers to the simultaneous substitution of ∅ for
both the variables members and telephones. The substitution is performed on
a description of the state of the system being specified. Whereas Z is a model-
based specification language, AMN is built on the idea of textual substitution.

When using the B-method to write a specification in AMN the specifier
needs to discharge certain proof obligations concerning the specification that
he or she is writing.4 One of these relates to the initialisation component of
the abstract machine and the formula that has to be proved is [T ] I, where T
is the initialisation and I is the invariant.5 In the specification being discussed
here this amounts to the following formula:

[members, telephones := ∅, ∅]
members ⊆ PEOPLE ∧
telephones ∈ PEOPLE ↔ PHONES ∧
dom(telephones) ⊆ members,

which simplifies to the formula

∅ ⊆ PEOPLE ∧ ∅ ∈ PEOPLE ↔ PHONES ∧ dom(∅) ⊆ ∅,

which is easily proved to be a theorem of set theory.
Another proof obligation relates to the invariant. It has to be shown that

the invariant is not trivially false. That is to say, it has to be shown that ∃x · I,
where x is the list of variables of the abstract machine and I is the invariant.5

In the case of the current specification this is the following formula:

∃members, telephones ·
members ⊆ PEOPLE ∧
telephones ∈ PEOPLE ↔ PHONES ∧
dom(telephones) ⊆ members,

which is easily seen not to be trivially false.

4 The Operations

The operation of successfully adding a member of the organisation to the
database is specified in Z by means of the schema AddMember:

4Note that whereas AMN is a language that is used to write specifications of software
systems, the expression ‘the B-method’ is used to refer to the way in which—according to
Abrial—AMN should be used to write such specifications.

5In the general case this proof obligation is more complicated because an abstract machine
may contain a constraints and a properties component. See [1] for more details.

AddMember
∆PhoneDB
name? : PEOPLE

name? /∈ members
members′ = members ∪ {name?}
telephones′ = telephones

Here, name? is the person who we are trying to add to the database. It is
possible in Z to obtain a precondition schema from any schema describing an
operation which makes use of the convention that before variables are unprimed
and after variables are primed. This is done by hiding or existentially quan-
tifying over the after and output variables. Doing this here—and simplifying
what is obtained—results in the schema PreAddMember:

PreAddMember
PhoneDB
name? : PEOPLE

name? /∈ members

There is no requirement when producing a specification written in Z to work
out what the precondition schema is.

In AMN the operation of making someone known to the database is specified
like this:

AddMember (name in) =̂

pre

name in ∈ PEOPLE ∧
name in /∈ members

then

members := members ∪ {name in}
end

A minor difference between Z and AMN is that AMN does not have the conven-
tion that input variables are decorated with a question mark. The operation
AddMember (name in) is defined to be a pre-conditioned substitution, that is
to say, it is a substitution that is only carried out if the formulas that occur
between the key words pre and then are satisfied. Note that if these formulas
are not satisfied, then anything might happen.

When the B-method is being used to write AMN specifications, then when-
ever an operation is specified there is a proof obligation to be carried out. This
involves proving that the invariant is preserved by the operation, that is to say,
the formula I ∧ Q ⇒ [S]I has to be established, where I is the invariant and

the operation is pre Q then S end.5 In this case this is the formula:

(members ⊆ PEOPLE ∧
telephones ∈ PEOPLE ↔ PHONES ∧
dom(telephones) ⊆ members ∧
name in ∈ PEOPLE ∧
name in /∈ members)⇒
([members := members ∪ {name in}]

members ⊆ PEOPLE ∧
telephones ∈ PEOPLE ↔ PHONES ∧
dom(telephones) ⊆ members).

The proof of this is straightforward—if a little tedious.
The specification of the operation labelled (2) in the informal description of

the system being specified—given in section 2 above—is straightforward and
introduces no new ideas.

As already mentioned, when using the B-method a proof obligation has to be
discharged whenever an operation is specified in AMN. When using Z, however,
there is no such obligation. Ensuring that the state invariant is preserved is
left entirely at the intuitive level. Thus, it would be possible to include in a Z
specification the following schema as an attempt to specify operation (3):

WrongAddEntry
∆PhoneDB
name? : PEOPLE
number? : PHONES

name? 7→ number? /∈ telephones
telephones′ = telephones ∪ {name? 7→ number?}
members′ = members

This is perfectly acceptable to fuzz, as fuzz is only a syntax and type checker
and so cannot be expected to catch this kind of error. However, if name? is not
an element of members, then the predicate-part of this schema is equivalent
to the constant formula false, because the state invariant dom(telephones) ⊆
members is not preserved: dom(telephones′) contains name? which is absent
from members′. In such a simple case it is not too difficult to see that some-
thing is not correct about WrongAddEntry, but in a large and complicated
specification it is easy to overlook such faults. The following is a correct version
of this operation:

AddEntry
∆PhoneDB
name?:Person
newnumber?:Phone

name? ∈ members
name? 7→ newnumber? 6∈ telephones
telephones ′ = telephones ∪ {name? 7→ newnumber?}
members ′ = members

It would be possible to specify an operation in the AMN language that is
incorrect in the same way that WrongAddEntry is incorrect, but by using
the B-method to develop specifications such an incorrect specification would
be quickly discovered because the relevant proof obligation could not be dis-
charged.

The specification of (4) is straightforward and introduces no new ideas.
The two operations (5) and (6) do, however, introduce some new ideas when
specified in AMN. Operation (5) is specified in Z by means of the schema
FindPhones:

FindPhones
ΞPhoneDB
name? : PEOPLE
numbers! : P(PHONES)

name? ∈ dom(telephones)
numbers! = telephones(|{name?}|)

In AMN this looks as follows:

numbers out←− FindPhones (name in) =̂

pre

name in ∈ dom(telephones)

then

numbers out := telephones[{name in}]
end

The novel feature introduced here is the way in which outputs are indicated in
AMN. They occur first, followed by a left-pointing arrow and then comes the
name of the operation being defined. Thus, the variable numbers out here is
the output of the operation FindPhones for the input name in. Two minor
differences between Z and AMN to notice here are that AMN does not have
the Z convention of indicating output variables by decorating them with an
exclamation mark and, secondly, that the syntax for the relational image of a
set through a relation is slightly different in the two notations.

The inclusion of the schema ΞPhoneDB in FindPhones tells us that that
operation does not involve any changes being made to the values of the state
variables. In the case of the AMN specification, this means that no substitutions
are made on the state variables.

Operation (6) is specified analogously to operation (5).

5 Completing the Specification in Z

The schema AddMember tells us what happens only if the precondition is sat-
isfied. In order to complete the specification so that something sensible happens
even when the precondition is not satisfied we need the schema AlreadyMember,
which is defined like this:

AlreadyMember
ΞPhoneDB
name? : PEOPLE
rep! : REPORTS

name? ∈ members
rep! = Already Member

If we try to add a name to the set members which is already present, then a
suitable error message is output. In order to provide some sort of indication
that the operation of adding a name to members has been successful we define
a schema Success:

Success
rep! : REPORTS

rep! = Okay

It is now possible to specify the total operation DoAddMember like this:

DoAddMember =̂AddMember ∧ Success
∨

AlreadyMember.

DoAddMember, unlike AddMember, is defined for all conceivable inputs. The
specification of the other total operations is similar, although further error
schemas need to be defined. The actual schemas involved in all this are un-
complicated and are summarised in the following table:

operation total successful error
specification outcome schemas

(1) DoAddMember AddMember AlreadyMember
(2) DoRemoveMember RemoveMember NotMember
(3) DoAddEntry AddEntry NotMember

EntryAlreadyExists
(4) DoRemoveEntry RemoveEntry UnknownEntry
(5) DoFindPhones F indPhones UnknownName
(6) DoFindNames FindNames UnknownNumber

6 Completing the Specification in AMN

It is not surprising that AddMember (name in), FindPhones (name in), and
the other AMN operations which describe successful operations are all placed
in the operations component of the abstract machine Phone. We also need to
decide where we are going to put the specification of errors. We have decided to
put them into Phone. It may seem strange at first sight to include operations
that output error messages in the abstract machine Phone, but the reason for
doing so is that these operations have the same invariant as do the other opera-
tions, like AddMember (name in) and FindPhones (name in). Schematically,
therefore, the machine Phone looks like this:



259

machine

Phone

sets

PEOPLE;PHONES;REPORTS = { Okay, . . . }
variables

members; telephones

invariant

members ⊆ PEOPLE ∧ telephones ∈ PEOPLE ↔ PHONES ∧
dom(telephones) ⊆ members

initialisation

members, telephones := ∅, ∅
operations

AddMember (name in) =̂ . . . ;

RemoveMember (name in) =̂ . . . ;

AddEntry (name in, number in) =̂ . . . ;

RemoveEntry (name in, number in) =̂ . . . ;

numbers out←− FindPhones (name in) =̂ . . . ;

names out←− FindNames (number in) =̂ . . . ;

rep out←− AlreadyMember (name in) =̂ . . . ;

rep out←− NotMember (name in) =̂ . . . ;

rep out←− EntryAlreadyExists (name in, number in) =̂ . . . ;

rep out←− UnknownEntry (name in, number in) =̂ . . . ;

rep out←− UnknownName (name in) =̂ . . . ;

rep out←− UnknownNumber (number in) =̂ . . .

end

The operation which just outputs a message saying that a successful outcome
has been achieved is placed in a machine called PhoneCtx. It is a different
kind of operation from the others encountered so far because it is stateless and
that, indeed, is the reason for putting it in a separate machine. One of the
most important parts of a machine is the description of the state, since the
operations describe substitutions to be performed on this, therefore it seems
sensible to distinguish between operations on this state and others, and to place
these two different kinds of operation in separate machines.

machine

PhoneCtx

uses

Phone

operations

rep out←− Success =̂ begin rep out := okay end

end

Note the presence here of a uses component in the machine PhoneCtx. This
is one of several ways in which one machine can make use of another one. It
means that the machine PhoneCtx can access the variables, sets and constants
introduced in the machine Phone, but it cannot alter the values of any of the
variables introduced in Phone. The constant Okay that is used in PhoneCtx
is introduced in the machine Phone.

In order to complete the specification of the Internal Telephone Number
Database in AMN we define the machine DoPhone as follows:

machine

DoPhone

includes

Phone, PhoneCtx

operations

rep out←− DoAddMember (name in) =̂ . . . ;

rep out←− DoRemoveMember (name in) =̂ . . . ;

rep out←− DoAddEntry (name in, number in) =̂ . . . ;

rep out←− DoRemoveEntry (name in, number in) =̂ . . . ;

rep out, numers out←− DoFindPhones (name in) =̂ . . . ;

rep out, names out←− DoFindNames (number in) =̂ . . .

end

Note the presence of an includes component in the machine DoPhone. This
means that that machine can access the variables, sets and constants introduced
in the machine Phone and PhoneCtx and, furthermore, it can also alter the
values of any variables introduced in those machines, but it can only do this
by calling the operations introduced in those machines. The reason for this
restriction is that it allows us to ensure that the invariants of the included
machines are maintained by any operations defined in the including machine.

DoAddMember (name in), DoRemoveMember (name in) and the other
total operations are all built up out of the corresponding partial operations and
appropriate error operations as you would expect. A single example should
suffice to explain how this is done:

rep out←− DoAddMember (name in) =̂

pre

name in ∈ PEOPLE

then

if name in /∈ members

then

AddMember (name in)||rep out←− Success

else

rep out←− AlreadyMember (name in)

end

end

The symbol || that occurs in this specification represents simultaneous substi-
tution. Note also the presence of the conditional substitution

if P then S else T end,

which has the effect that the substitution S is performed if P is true, but T is
carried out if P is false.

7 Conclusion

In this paper we have compared some aspects of the formal specification lan-
guages AMN and Z and the main differences that we have found between them
will be summarised here:6

(1) This first point relates to the large-scale organisation of specifications.
(As refinement has not been mentioned above the following remarks do
not apply to it.) There is a precise description of the syntax of Z—found,
for example, in [10], pp. 142–146—and according to this a Specification
is a well-defined unit. There are no ways, however, of combining two
specifications—except that of concatenating them.7 (In doing this it
may also be necessary to rearrange the order of some of the specifica-
tion components and to remove duplicate items.) In AMN, by contrast,
an abstract machine is a well-defined unit and there are several carefully
thought-out ways of combining machines. Two of these have been illus-
trated in this paper. We have shown, for example, the way in which one
machine uses another and also the way in which one machine includes
other machines. There are still further ways of relating machines to one
another.

(2) Z is based on first-order logic and set theory and many people who advo-
cate the use of Z say that one of the advantages of using a mathematical
notation is that it is possible to prove things about a formal specifica-
tion, but not much guidance is given to the specifier to help him or her
decide which properties of a specification are worth proving and which
are not. In the B-method, however, certain properties of a specification
are singled out as proof obligations and if these are carried out, then the
specifier knows that the specification he or she has written has a number
of desirable properties. In this paper, for example, we have mentioned
the following proof obligations:

(a) that the substitution in the initialisation component of an abstract
machine establishes the invariant;

(b) that the invariant is not trivially false; and

(c) that each operation in an abstract machine preserves the invariant.8

6Further comparisons between Z and AMN—some of which overlap with those contained
here—can be found in [5].

7This comment refers to standard Z as defined in Spivey’s book. Much research is currently
being done to see how object-oriented features can be added to Z. See, for example, [6] and
[2].

8When using Z there is no need to prove that the invariant is maintained and—as shown
above—this may cause problems.



262

Further proof obligations need to be discharged when more complicated
things are being done. This is especially useful, for example, in a non-
academic environment where there may be various pressures at work and
in such circumstances it may be useful to know that all necessary proof
obligations have been carried out.

(3) In specifying an operation in Z by means of a schema you need to explicitly
mention whether or not a variable is affected by means of the operation.
For example, in the schema AddMember above we have had to include
the formula telephones′ = telephones to indicate that this operation does
not change the value that the variable telephones has before and after
the operation has been carried out. When a large specification is being
written with a large number of variables it becomes tedious to have to
include all variables whose values are left unchanged by an operation.
In AMN, by contrast, there is no need to mention variables that are
unchanged by an operation.

(4) The preconditions of an operation specified in AMN must be explicitly
mentioned, whereas in a Z schema the preconditions are not singled out
in any way. It is possible, however, to calculate the precondition schema
from any schema used to specify an operation.

There are also further differences between Z and AMN that have not appeared
in the small example considered in this paper. Some of the most important of
these will be mentioned here:

(5) Formal specifications of real-life systems tend to become large and they
are difficult to manipulate in an error-free way by hand. It is highly
desirable, therefore, to have some sort of machine support to help with
the manipulation of specifications. In the case of Z the provision of such
machine support is in an early stage of development. Some of the existing
tools are: fuzz (a syntax and type checker which can also typeset Z
specifications), CADiZ (a suite of tools whose functionality is similar to
that of fuzz), ProofPower and Zola (both of which are claimed to be able
to carry out some proofs about Z specifications). With AMN, however,
the situation is very different. There exist several sophisticated pieces
of software, namely the B-Tool and the B-Toolkit, which greatly help
with the production of specifications written in AMN. Moreover, if the
B-method is being used, then given a specification written in AMN the B-
Toolkit will generate all the proof obligations that need to be discharged
with respect to it and it is able to discharge many of these automatically.
A small amount of human assistance may be needed for a complicated
proof obligation to be discharged.

(6) There is currently no universally accepted way in the Z community of
relating a low-level specification to program code, though several ap-
proaches are being investigated. (For some of these, see [3], [7],9 and
[11]. This is not meant to be an exhaustive list.) The situation is very
different in the case of AMN. Here a low-level specification is related to
program code by means of Dijkstra’s weakest-precondition calculus and

9This paper relates Z and the refinement calculus expounded in [8].

this is integrated into the entire specification-design-implementation pro-
cess associated with the AMN language.

We draw no conclusions as to whether Z or AMN is the better specification
language. This question really does not have a clear sense. Specification lan-
guages are complicated systems with large numbers of properties and it is
possible to compare them in many different ways and along several dimensions.
It is ridiculous to attempt to give a single, unified, global rating to either Z or
AMN. The purpose of this paper is to draw the reader’s attention to AMN,
which—we believe—is not as well known as it should be and especially to its
modularisation features.

